

Setting the Table... of Contents: Using SAS ODS PDF Features to Organize, Link, and Navigate a Report

Betsy Enstrom, IDEaS-A SAS Company, Bloomington, MN

ABSTRACT

Today, very few reports are actually printed this corresponds to the growing dependence on electronic browsing, researching, and reporting. When documents are viewed electronically, they are often viewed out of order. This creates a need for well-organized reports that are easily navigated by means such as table of contents, hyperlinks, and bookmarks rather than relying on the standard page up and down. This is done with the clever combination and use of several ODS PDF options such as ANCHOR, PROCLABEL, PDFTOC combined with the COMPUTE block in the REPORT procedure. You can set the table of contents with hyperlinks and make use of the bookmark structure already in PDFs. This saves readers time and frustration while navigating your report.

INTRODUCTION

This paper describes how to create an easy to navigate report that is well organized. It makes use of several ODS and procedure options to create a PDF report that is both well-organized and descriptive. It allows the audience to find results much quicker than the usual Page Up, Page Down, Home, and End keys. By combining clear titles and descriptions with ODS options like ANCHOR, PROCLABEL, PDFTOC, and TEXT the report is a well-designed set of analysis that allows the audience to focus on the analytics rather than thumbing through the pages.

FORKS AND PLATES TO SETTING THE TABLE

Consider Figure 1, notice the bookmarks pane on the left of the screen and the Table of Contents (TOC) in the center of the page. The viewing options of the bookmark page are available under View > Show/Hide > Navigation Pane > Bookmarks. This allows the reader to view or hide the bookmarks. The TOC is colorful, well-spaced, and contains active URL links to the tables and figures within the report.

Figure 1: Table of Contents and Bookmark Pane

Table of Contents	
Hotel Level Graphs - Basic	
Occupancy, ADR, and RevPAR - Daily	Occupancy, ADR, RevPar, and respective 28-day moving averages by occupancy date
Occupancy, ADR, and RevPAR - Monthly	Monthly averages of occupancy, ADR, and RevPAR
Distribution of Rooms Sold by Day of Week	Percentage of total rooms sold by day of week with corresponding ADR and RevPAR
Market Segment Level Graphs	
Market Segment Distribution and Average Daily Rate - Top 15	The number of rooms sold and the average daily rates for the top 15 market segment contributors
Market Segment Production - Top Six	The number of rooms sold each month for the top six market segment contributors
Hotel Level Graphs - Additional	
Number of Out-of-Order Rooms by Occupancy Date	Capacity, out-of-order, and 28-day moving average of out-of-order rooms by occupancy date
Occupancy by Occupancy Date with Out of Order	Occupancy, 28-day moving occupancy average, and occupancy with out-of-order by occupancy date
Hotel Performance by Level of Rooms Sold	Occupancy percentage and ADR arranged in ascending occupancy order with linear trends of ADR and RevPAR
Room Night Categorization	Occupancy level, ADR, busy and shoulder nights by occupancy date
Hotel Performance by Night Type	Distribution of nights, ADR, and RevPAR by night type

It is clear from the report that there are four main areas where SAS can be leveraged to do the heavy lifting of the organization. The ODS PDF ANCHOR option creates a reference point and linkable sections in your analysis or report. The ODS PROCLABEL option controls what is displayed in the first branch of the Bookmarks pane. The ODS PDFTOC option controls how many levels the Bookmarks pane contains. In this report, the PDFTOC is set to two. The COMPUTE block in the REPORT procedure allows the user to create an input data set that contains the anchors, procedure titles, and descriptions to create an electronic TOC.

ANCHORS DOWN

In order to create reference points and linkable sections of your report, use the ODS PDF ANCHOR. This allows the reader to select a hyperlink and go directly to the desired procedure output. The actual anchor `plotb` will be used as the value of the `link` variable in the input data set for the REPORT procedure TOC.

```
ods pdf anchor = 'plotb' startpage=now;
proc sgplot data=oar_daily
  description="Occupancy, ADR, RevPar, and their respective
  28-day moving averages by occupancy date";
  title4 'Occupancy, ADR, and RevPAR - Daily';
...
run;
```

In the sample code above, the ODS PDF ANCHOR is declared right before the procedure it is referencing.

LABELING AND DESCRIBING YOUR PROCS

To help in adding a narrative and organizing your report, use the ODS PROCLABEL option to label a procedure. What better label than the title of your report so here, the PROCLABEL is the title of a report. This allows the reader to easily select a report item directly from the Bookmarks pane in the PDF. With PROCLABEL you will leverage the value of the `titles` variable in the input data set for the REPORT procedure TOC. The sample code below should be placed before the procedure output you wish to label. Note, it is simply the title of the procedure it is labeling.

```
ods proclabel "Occupancy, ADR, and RevPAR - Daily";
```

In an effort to describe your analysis or results, use the CONTENTS or the DESCRIPTION option on the PROC statement. Here, the CONTENTS or DESCRIPTION allows you to provide more detail of the results than would look more aesthetically pleasing in a title. The CONTENTS or DESCRIPTION will be the value of the `caption` variable in the input data set for the REPORT procedure TOC. In the sample code below, the DESCRIPTION option is used in the SGPLOT procedure. The REPORT procedure uses the CONTENTS option in place of the DESCRIPTION option. Check what the corresponding option is in your procedure.

```
proc sgplot data=oar_daily
  description="Occupancy, ADR, RevPar, and their respective 28-day
  moving averages by occupancy date";
  title4 'Occupancy, ADR, and RevPAR - Daily';
  ...
run;
```

In addition to describing your results in the second level of the Bookmarks pane in the PDF, add the same text in the DESCRIPTION or CONTENTS option to your report below the graph or table with the ODS PDF TEXT option. This allows the reader to easily interpret a bit more insight as to what the results are reporting.

```
ods pdf text= '^S={font_size=12pt}';
ods pdf text = 'Occupancy, ADR, RevPar, and their respective 28-day
  moving averages by occupancy date';
```

TRIMMING YOUR (TOC) TREES

Certain procedures have many levels of output and hence many levels of branches of a tree in the Bookmark pane. In order to control how many levels of the procedure output are displayed in Bookmark pane and focus your reader's attention, use the ODS PDFTOC.

```
ods pdf pdftoc = 2 file = "MWSUG_hotel.pdf";
```

ORGANIZE THE PLACE SETTINGS...AT YOUR TABLE

You have seen all the pieces, now it is time to see how they all come together. Finish the TOC input data set, called `titles2` in the sample code below, and display it all with the REPORT procedure. You have seen the `link`, `titles`, and `caption` variable previous sample report code. Now see the items come together in a DATA step. First, set the lengths of the input variables with the LENGTH statement as some of the `titles` and `caption` variable can be very long. Next, add a `level` variable to create nice break in the sections or chapters of the report in the TOC. For each item displayed in the report, create a `link`, `level`, `titles`, and `caption` followed by an OUTPUT statement. The sample code below illustrates how to create the input data set.

```
data titles2;
  length link $8 level $50 titles $300 caption $400;

  link    = "plotb";
  level   = "Hotel Level Graphs - Basic";
  titles  = "Occupancy, ADR, and RevPAR - Daily";
  caption = "Occupancy, ADR, RevPar, and respective 28-day moving averages by
  occupancy date";
  output;
  link    = "plotc";
  level   = "Hotel Level Graphs - Basic";
  titles  = "Occupancy, ADR, and RevPAR - Monthly";
  caption = "Monthly averages of occupancy, ADR, and RevPAR";
  output;
  ...
run;
```

The REPORT procedure's many formatting options makes for a nice looking and well-functioning TOC. First, declare the PROC statement options like NOHEADER, NOWD, SPLIT, and of course, the CONTENTS option.

```
proc report data = titles2 nowd split = ' ' noheader
  contents = 'Table of Contents with Descriptions';
```

Next, add the COLUMN statement with the designated variables.

```
column level link titles caption;
```

Then, add all the DEFINE statements for the variables with the background, foreground, and font options that you desire.

The `link` variable is very important even though it is not printed with the NOPRINT option on the DEFINE statement.

```
define link          / noprint;
```

Use the `level` variable to add a BREAK statement. This will make the TOC better spaced easier to read.

```
break after level    / skip skip;
compute before level/ style=[just=l background = cx1495d3
  font_face   = '<ttf> Century Gothic' font_size   = 12pt
  foreground  = white font_weight = bold];
  line level $35.;
endcomp;
```

Next, it is time to see some of the magic the COMPUTE block in the REPORT procedure has to offer. Create a COMPUTE block with the `titles` variable. The COMPUTE block uses the `link` variable to turn the titles into active hyperlinks to the desired section of the report.

```
compute titles;
  rtag = "#"||trim(link);
  call define(_col_, 'url', rtag);
endcomp;
```

Use the CALL DEFINE function to turn the trimmed `link` into an active URL.

CONCLUSION

It is clear to see, that nothing has been too advanced or complicated in the creation of a nice TOC. Simply make use of ODF PDF, ODS, and PROC statement options already in software. It becomes easy to build sophisticated reports that work well both electronic and printed. Your audience will thank you for creating easy to navigate and narrative report so they may focus on your results.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Betsy Enstrom
IDeaS-A SAS Company
8500 Normandale Lake Blvd. Ste. 1200
Bloomington, MN 55437
Work Phone: (952)698-4205
Fax: (952)698-4299
E-mail: betsy.enstrom@sas.com
www.ideas.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.